

SE Field Article: ProfileUnity Boot Environment Variables

Last Updated: 05/25/2017

Document Purpose: Show how to create environment variables at boot time that can be used by ProfileUnity.

Script Use Case:

ProfileUnity Portability, Folder Redirection and other modules can use environment variables for many functions.

This document provides an example on how to create an environment variable at boot time based on part of the machine name. This environment variable can then be used by any of the ProfileUnity modules to change the behavior of the rule based on parts of the machine name.

One of the reasons you would want to do this is to simplify and reduce the number of rules needed to be created in the Portability or Folder Redirection modules.

Example: ATL - HQ

Machine Name Examples: ATL010-vm1, ATL010-vm2...ATL010-vm250

Machine Names All Start with ATL010

Output would be: "ATL010"

Script Function:

1. Determine Data Center Location - (USA or UK)
 - Determined by looking for "UK" in machine Name
 - ***This can be modified for your needs***
 - Used to set UNC Base Mapping
2. Get environment variable "COMPUTERNAME"
3. Return all characters from computer name up to first instance of "-"
 - *** This can be modified for your needs***
4. Create New Environment Variables to be used by ProfileUnity

Script Notes:

1. This script is not supported by Liquidware.
 - Reason: You can modify scripts to do an infinite number of things. Liquidware cannot support this.
2. This script should only be used in environments where the computer name is controllable and predictable.
3. This script has been tested on:
 - Windows 7 & 10
 - 2008R2 and 2012R2 in "Desktop Mode"
 - DO NOT USE on a Multi-User OS!!!**

Script Deployment:


Script Location:

This script should be in a location that is accessible to all users in a “Read-Execute” only mode. Normally it would be placed on the domain controller netlogon directory under “Scripts”.

Example: \\YourDomain\NETLOGON\Scripts

Startup GPO:

You can use the existing ProfileUnity GPO if you like. This is a computer “Startup” script so it is only executed when the machine boots.




Note: Order of Execution does not matter.

Note: If you have multiple domains, the settings above only need to be applied to the Domain OU that the computer boots in.

ProfileUnity Usage Example:

If you have multiple clients (MSP's), datacenters or sites and need to simplify your folder redirection and portability rules based on machine name "Contains" XYZ this is the easiest way.

Folder Redirection Example:

Edit Folder Redirection Setting

Filter

Windows 7

No comments

Description

Shell Folder

Desktop

☐ Reset to Default

Redirect to Folder REQUIRED

\\%MachineNameMap%\%username%\Desktop

FULL Batch File Script:

```

::Script Name = BootEnvVar.bat
::Script Last Modified: 01/19/2017
::Revision History:
:: ID      Date      Notes
:: cwalker 10/20/2016 First release to general public after testing with MSP client
:: cwalker 01/19/2017 Documentation Updates

:: ----- Start BootEnvVar.bat-----
::----- Define Script Variables - Start -----
:: UNC Base Path to Store User Profile Information
:: Note: User must have access to directory
:: Note: A hidden directory is recommended - Use "$" at the end of the share name
:: Profile Storage Location Definition
:: US Data Center - Profile Storage Path
SET USA_Path_s=MyDomain.com\Server\ShareName$\
:: UK Data Center - Profile Storage Path
SET UK_Path_s=MyDomain.com\UKServer\ShareName$\
:: Default Base Path is US Data Center
SET UNC_Base_Path_s=%USA_Path_s%
::----- Define Script Variables - End -----

:: If UK Machine - Change UNC_Base_Path_s - If Not then SkipUK
:: - (Checks Environment Variable COMPUTERNAME contains "UK")
IF x%COMPUTERNAME:UK=%==x%COMPUTERNAME% GOTO SkipUK
:: Set Base Path for UK Machine - Since ATL was not found in ComputerName Variable
SET UNC_Base_Path_s=%UK_Path_s%
:SkipUK

:: Get %COMPUTERNAME% Environment Variable and split it to get Common Name (Common Name is Before First
".")
for /f "tokens=1 delims=- " %%a in ("%computername%") do SET ComputerCommonName_s=%%a

:: Set Environment Variables for Machine to use for ProfileUnity
:: SETX for Permanent Environment Variable
:: - /M option is for HKLM - Since this is ran at boot time
:: In this example a new environment variable will be created called "MachineNameMap"
SETX MachineNameMap "%UNC_Base_Path_s%%ComputerCommonName_s%" /M

:: No Need to Clean up SET Variables - They go away after exit of CMD Shell
:: Exit Shell
exit
:: Script Processing - End
:: ----- end BootEnvVar.bat-----

```